

1 Heurísticas

Nos métodos de busca vistos até agora, a ordem de variáveis adotada é a ordem apresentada na lista e a ordem de valores para cada variável é fixa. Para problemas maiores, pode ser mais conveniente considerarmos ordenações diferentes para cada tipo de problema. Essas ordenações são geralmente escolhidas por alguma heurística e, por isso, são chamadas ordenações heurísticas. Existem, naturalmente, duas classes de ordenações heurísticas: ordenação heurística de variáveis e de valores.

Antes de prosseguirmos, será útil falarmos sobre vetores, que serão utilizados nos programas a seguir.

1.1 Vetores

Listas são interessantes, mas, eventualmente, podemos querer realizar acesso a em tempo constante a um membro qualquer, o que não é possível. Em outras linguagens, esse tipo de acesso é por costume realizado através de vetores. Vetores existem em Prolog padrão, de certa forma. Como os argumentos de funtores podem ser acessados em tempo constante por meio do funtor `arg/3`, a princípio, um funtor qualquer poderia ser usado como um vetor, o acesso aos membros do qual sendo feitos por meio de `arg/3`. Essa abordagem, no entanto, não só não parece muito elegante, como é pouco prática, . Se temos algo que chamamos de vetor, gostaríamos de poder fazer algo como `Vetor[2] =/= 4`, por exemplo, o que com os meios do Prolog padrão não é possível.

Para esse tipo de situação, no *ECLⁱPS^e* existe uma implementação de vetores como um açúcar sintático semelhante ao que vimos na implementação de listas, permitindo justamente esse tipo de utilização. Em particular, um vetor, é escrito como uma variável seguida do funtor de lista (como em `Var[4]`), ou, se anônimo, como `[](A, B, C)`. Note que, assim, a representação de um vetor ou de uma lista vazia é a mesma. Para criar “vetores”, possivelmente multidimensionais (na realidade, matrizes), existe o predicado `dim/2`, que pode ser usado ou para criar vetores ou para extrair sua dimensão.

1.2 As *n*-rainhas

Analisaremos o efeito de algumas heurísticas pela resolução do *problema das n-rainhas* quando $n > 3$ ¹. É um problema por restrições tão bem conhecido que um estudante de programação por restrições poderia (talvez com razão) se sentir enganado se terminasse um livro sobre o tema sem saber como resolver esse problema em especial. A discussão a seguir é baseada, em grande parte, em [1].

¹Quando $n = 2$ ou $n = 3$, o problema não tem solução

Para quem não conhece, o *problema das n -rainhas* é o problema de alocar n rainhas em um tabuleiro como o de xadrez, mas $n \times n$, com a restrição de que as rainhas não têm permissão para se atacarem segundo as regras usuais de xadrez.

Existe um algoritmo polinomial para a resolução desse problema (que não veremos aqui), mas o que veremos muitas vezes é mais conveniente e até mais rápido. Antes de considerarmos o papel das heurísticas, vejamos como é uma solução mais simples. Notamos primeiro que, como temos n rainhas, precisa existir uma rainha em cada linha e em cada coluna. Ou seja, para cada coluna (ou cada linha) só precisamos indicar em qual linha (coluna) a rainha está, o que significa que a solução requer uma lista² de n posições, não uma matriz (como alguém poderia supor a princípio). O domínio de cada posição na lista é o intervalo inteiro $[1, n]$ e as restrições são que só pode existir uma rainha por linha (ou por coluna), o que se traduz na restrição `alldifferent(Cols)`, em que *Cols* é o vetor de colunas. Ademais, só pode existir uma rainha por linha, o que se traduz nas restrições $X_i^3 - X_j \setminus = \setminus \text{I} - \text{J}$ e $X_i - X_j \setminus = \setminus \text{J} - \text{I}$, em que X_i e X_j são membros distintos da lista de colunas (ou linhas).

²Ou um vetor.

³Por conveniência, usamos X_i no lugar de $X[i]$.

```

%% rainhas(-Rainhas, ++Numero)
% Rainhas contém a posição de cada rainha por coluna
%

:- lib(ic)

rainhas(Rainhas, Numero) :-
    dim(RainhaStruct,[Numero]),
    restricoes(RainhaStruct, Numero),
    struct_para_lista(RainhaStruct, Rainhas),
    busca(Rainhas).

restricoes(RainhaStruct, Numero) :-
    ( for(I,1,Numero),
      param(RainhaStruct,Numero)
    do
      RainhaStruct[I] :: 1..Numero,
      (for(J,1,I-1),
        param(I,RainhaStruct)
      do
        RainhaStruct[I] #\= RainhaStruct[J],
        RainhaStruct[I]-RainhaStruct[J] #\= I-J,
        RainhaStruct[I]-RainhaStruct[J] #\= J-I
      )
    ).

struct_para_lista(Struct, Lista) :-
    ( foreacharg(Arg,Struct),
      foreach(Var,Lista)
    do
      Var = Arg
    ).

```

Código 1: Queens

Esse programa primeiro declara o vetor *RainhaStruct*, impõe a ele as restrições mencionadas, transforma o vetor em lista e realiza a busca (como explicada no Capítulo 10). O programa, no computador do autor, consegue resolver o problema por esse programa para N até 29, mas, adotando um tempo limite de 50 segundos, $N = 30$ já está fora de mão.

Alguém poderia dizer que isso ocorre porque esse código não foi feito de maneira inteligente, já que a assimetria do problema, presente no fato de que a alocação de uma rainha em uma coluna central, por exemplo, resulta em maiores restrições nas posições das outras rainhas, não está representada no código. Esse realmente é o caso e, para fazê-lo deixar de ser, criamos uma nova versão do

programa.

1.3 Ordenação de variáveis

A alteração consiste na adição de um argumento *Heur* (de heurística) no predicado *rainhas/2* (que vira *rainhas/3*). Esse predicado nos diz qual a heurística usada para ordenar as variáveis. Podemos definir o *rainhas/2*, do exemplo passado, como um *rainhas/3* no qual o último argumento é “naive” (indicando que a heurística usada é, por assim dizer, boba).

```
busca(Rainhas, naive) :- labeling(Rainhas).
```

Código 2: Heurística Ingênua

Uma heurística mais esperta nesta situação seria começar a testar as variáveis do meio. Chamaremos essa heurística de *middle_out*: ela faz com que a busca comece do meio e vá alternando entre os vizinhos do lado esquerdo e direito, a partir do meio.

```
:- lib(lists).

busca(Rainhas, middle_out) :-
    middle_out(Rainhas, RainhasDeSaida),
    labeling(RainhasDeSaida).

middle_out(Lista, ListaDeSaida) :-
    halve(Lista, PrimeiraMetade, UltimaMetade),
    reverse(PrimeiraMetade, RevPrimeiraMetade),
    splice(UltimaMetade, RevPrimeiraMetade, ListaDeSaida).
```

Código 3: Heurística Meio

Aqui usamos a biblioteca *lists*, que contém, entre outros predicados, *halve/3*, *reverse/2* e *splice/3*. O que o primeiro e o segundo desses predicados faz é claro. O que o terceiro faz é juntar em *ListaDeSaida* as *UltimaMetade* e *RevPrimeiraMetade* colocando um membro de cada alternadamente (como uma função *merge* do *merge sort*).

À parte dessa heurística, existe a *first_fail*, que seleciona como a próxima variável a que tem menos valores restantes no domínio. Para implementar essa heurística, usaremos o predicado da biblioteca *ic delete(-X, +List, -R, ++Arg, ++Select)*, que remove uma variável *X* da lista de variáveis *List*, deixando o resultado em *R*. O parâmetro *Arg* indica se a lista é uma lista de fato ou um funtor (no segundo caso, os argumentos do funtor são tratados como se fossem elementos de uma lista), e o *Select* indica o parâmetro de seleção (uma lista dos parâmetros possíveis pode ser encontrada na documentação do

ECL^iPS^e). No caso, nosso parâmetro *Select* é *first_fail*. Nossa busca com *first_fail* fica da seguinte forma:

```
busca(Lista, first_fail) :-
  ( fromto(Lista, Vars, Resto, [])
  do
    delete(Var, Vars, Resto, 0, first_fail),
    indomain(Var)
  ).
```

Código 4: Heurística First Fail

o parâmetro 0 indica que de fato lidamos com uma lista.

A experiência prática (que você pode realizar) indica que, para instâncias pequenas (para N pequenos), a diferença entre usar o *first_fail* ou não é pequena. Para N grandes, entretanto, a diferença é notável.

A priori, não temos motivos para não usar o *first_fail* e o *middle_out* juntos. Chamaremos essa heurística de *moff*:

```
busca(Lista, moff) :-
  middle_out(Lista, ListaDeSaida),
  ( fromto(ListaDeSaida, Vars, Resto, [])
  do
    delete(Var, Vars, Resto, 0, first_fail),
    indomain(Var)
  ).
```

Código 5: Heurística Meio + First Fail

1.4 Heurísticas de ordenamento de valor

A mesma observação usada para a heurística *middle_out* vale para a ordem dos valores escolhidos: a escolha de valores próximos ao centro restringem mais os valores de outras variáveis e têm a chance de falhar mais cedo. O ECL^iPS^e oferece um predicado *indomain/2* que nos ajuda nisso: seu segundo argumento nos dá um certo controle sobre a ordem em que os valores são testados. Em particular, *indomain(Var, middle)* começa a enumeração das variáveis pelo meio do domínio de *Var*. Existem também *indomain(Vars, min)* e *indomain(Vars, max)* que começam pelos menores valores e maiores, respectivamente (usando a ordem do domínio, pela qual “menor” é o que aparece antes na lista).

Podemos assim combinar nossa heurística anterior (*moff*) com esta (que chamaremos de *moffmo*):

```

busca(Lista, moffmo) :-
    middle_out(Lista, ListaDeSaida),
    ( fromto(ListaDeSaida, Vars, Resto, [])
    do
        delete(Var, Vars, Resto, 0, first_fail),
        indomain(Var)
    ).

```

Código 6: Heurística Meio First Fail Meio

Na tabela a seguir consta a quantidade de *backtrackings* por heurística para alguns valores de N :

Figura 1: Retirado de [1]

	naive	middle_out	first_fail	moff	moffmo
8-queens	10	0	10	0	3
16-queens	542	17	3	0	3
32-queens	—	—	4	1	7
75-queens	—	—	818	719	0
120-queens	—	—	—	—	0

1.5 Outras Considerações

O problema das n -rainhas é simétrico em que se uma solução tem a i -ésima rainha no i -ésimo quadrado da coluna i , outra solução teria a i -ésima rainha no i -ésimo quadrado da linha i . Perceba que uma solução é equivalente à outra, a menos de rotações de 90° do tabuleiro. Assim, quando a heurística de ordenação de variáveis *middle_out* sucede sem *backtrackings*, a heurística de ordenação de valores *indomain(Var, middle)* também deveria. Percebendo isso, podemos criar uma heurística de ordenação de valores que faz *backtrackings* como a *middle_out*. Depois de aplicada, uma “rotação” deve nos mostrar as mesmas soluções e na mesma ordem.

Para tanto, fazemos pequenas modificações no nosso programa. No lugar de iterar sobre a lista de variáveis, iteraremos sobre os valores do domínio. Ademais, no lugar de selecionar um valor para a variável atual (pelo *indomain/1*), selecionaremos uma variável para um valor (pelo *member/2*). Por fim, “rodamos” o resultado, para que os resultados produzidos sejam os mesmos produzidos pelo *middle_out*. Um código representando o que foi discutido é como segue:

```

rainhas(rotate, Rainhas, Numero) :-
    dim(RainhaStruct, [Numero]),
    constraints(RainhaStruct, Numero),
    struct_para_lista(RainhaStruct, QList),
    busca(QLista, rotate),
    rotate(RainhaStruct, Rainhas).

busca(QLista, rotate) :-
    middle_out_dom(QLista, MOutDom),
    ( foreach(Val, MOutDom),
      param(QLista)
    do
      member(Val, QList)
    ).

middle_out_dom([Q | _], MOutDom) :-
    get_domain_as_lista(Q, OrigDom),
    middle_out(OrigDom, MOutDom).

rotate(RainhaStruct, Rainhas) :-
    dim(RainhaStruct, [N]),
    dim(RRainhas, [N]),
    ( foreachelem(Q, RainhaStruct, [I]),
      param(RRainhas)
    do
      subscript(RRainhas, [Q], I)
    ),
    struct_para_lista(RRainhas, Rainhas).

```

Código 7: Heurística Roda

Curiosamente, os resultados para o *rotate* são muito melhores do que para *middle_out*, como pode ser visto na seguinte tabela, mostrando a quantidade de *backtrackings* por N e por heurística:

Figura 2: Retirado de [1]

	middle_out	rotate
8-queens	0	0
16-queens	17	0
24-queens	194	12
27-queens	1161	18
29-queens	25120	136

Isso ocorre porque o comportamento da propagação de restrições não é o mesmo para linhas e colunas. Se os valores para uma coluna são excluídos, exceto um, essa variável é instanciada ao valor restante. Esse é o comportamento exibido por *rotate*.

No entanto, no comportamento exibido por *middle_out*, se todos os valores de uma linha são excluídos exceto um, a propagação de restrições não instancia a variável. Isso ocorre porque as variáveis são representadas por colunas.

Uma forma de conseguir a mesma quantidade de propagação é usar uma redundância na formulação do problema. Neste caso, poderíamos adicionar outro tabuleiro ligado ao primeiro de modo que, se uma rainha *i* é posta no quadrado *j* de um tabuleiro, no outro, uma rainha *j* é posta no quadrado *i*. Técnicas desse tipo se mostraram muito eficientes.

1.6 O Predicado Search

Espero que a leitora não fique triste ao saber que grande parte do trabalho desenvolvido até agora poderia ser substituído pelo uso do predicado `search/6`. Ele tem 6 argumentos: `search(+Lista, ++Arg, ++Select, +Escolha, ++Metodo, +Opcao)`. O primeiro argumento, *Lista*, é uma lista de variáveis de decisão (as variáveis do problema) quando *Arg* é 0, ou uma lista de funtores compostos quando *Arg* > 0. Assumiremos que *Arg* é 0. O argumento *Select* representa uma heurística (feita pelo usuário ou não) de ordenação de variáveis, enquanto que, o *Escolha*, uma de ordenação de valores. Pelo *Metodo*, podemos selecionar a “forma” pela qual a busca deve ser feita: algumas opções são *complete* (que realiza uma busca por todos os valores), e *sbd*s (que faz uso da biblioteca SBDS de quebra de simetria para excluir partes simétricas da árvore de busca)⁴. O argumento *Opcao* é usado para passar parâmetros adicionais⁵, algumas das quais são *node(daVinci)*, para criar um desenho da árvore de busca usando a ferramenta de desenho daVinci, e *backtracks(-N)*, que nos dá em *N* a quantidade de *backtrackings* que ocorreram durante a busca.

⁴Recomendamos a consulta ao manual do *ECLⁱPS^e* / para mais detalhes sobre esses e outros métodos de busca.

⁵Novamente, é recomendado checar o manual do *ECLⁱPS^e* / para mais informações.

Leituras adicionais

- [1] Krzysztof R. Apt and Mark Wallace “Constraint Logic Programming using ECLiPSe” Cambridge University Press, The Edinburgh Building, Cambridge CB2 8RU, UK