

# 1 O Problema do General Bárbaro

Para ver se realmente compreendemos as ideias que vimos anteriormente, é sempre útil ver se conseguimos aplicá-las a algo concreto. Para isso, desenvolveremos aqui um problema específico (o “problema do general bárbaro”<sup>1</sup>). Mas faremos isso não com vistas em colocá-lo diretamente no *framework* desenvolvido até agora, mas sim em entendê-lo, pensar em diversos métodos de resolvê-lo e comparar esses métodos.

## 1.1 O Problema

O general romano Sulla, em suas campanhas pela Grécia, fez (entre outras coisas) o que qualquer outro general bárbaro faria em seu lugar, isto é, saqueou tesouros locais para obter vantagem na guerra (diferente de outros generais bárbaros, Sulla prometeu devolver parte do saque). Mas isso constituiu um problema, já que ele tinha uma grande variedade de tesouros para escolher levar ou não, enquanto que a capacidade de carga de seu exército era muito limitada, umas vez que dispunha de (relativamente) poucos navios. Mas o que criava a maior dificuldade nesse projeto era não o volume dos tesouros, mas o seu peso.

O valor de cada peça de tesouro é dado por uma função complicada de interesses internos e externos (no que influem, entre outros fatores, o preço pelo qual Sulla poderia vender dado item, no valor que cada soldado atribui ao item, no valor que seus conterrâneos atribuiriam ao item e na reputação que carregar um item lhe conferiria) e é sumariada no que chamamos de *valor Sulla*,  $\mathcal{S}$ , de um item. Depois de calcular o  $\mathcal{S}$  de cada item relevante, Sulla compilou a seguinte tabela preliminar com o peso médio e valor médio de cada item<sup>2</sup>:

Item	Valor	Peso	Quantidade
Livros	10	5	1000
Barras de Ouro	100	500	85
Estátuas de Ouro	500	4000	7
Jóias	50	50	200

Sem muito sucesso em resolver o problema, Sulla mandou que chamassem um matemático local para resolver o problema. Theon de Smyrna, estava por perto e foi convocado. No início, se sentiu incomodado no por ter que deixar de lado seu trabalho com os sólidos de Platão em favor de um ajudar um bárbaro a resolver um problema desse tipo inferior, como considerava, mas, notando que pode ser mais difícil para um morto desenvolver trabalhos com os sólidos de Platão, decidiu ajudar Sulla com seu problema.

<sup>1</sup>Também conhecido como “o problema da mochila”, ou *knapsack problem*, em inglês.

<sup>2</sup>Não há evidências de que tal tabela tenha sido feitas, mas é mantida aqui pela conveniência ao exemplo.

Ao ter a situação explicada, Theon deu uma leve risada e disse o que se segue. “Você me dá um problema grandioso e difícil, oh grande Sulla! Os Deuses devem ter nos posto juntos, porque eu devo ser um dos poucos helenos, se não o único, que sabe resolvê-lo! Isto porque vi meu pai o resolvendo quando era criança. Espero que não se incomode em ouvir por alguns momentos um velho como eu relembrar a infância, ainda mais que isso te será útil.”

“Continue.” - disse Sulla.

“Meu pai era um fazendeiro muito forte, mas também tinha o seus limites. Às vezes, precisava realizar trabalhos em partes distantes e, para isso, precisava levar suas ferramentas. Cada ferramenta faria o seu trabalho mais fácil, ou o ajudaria a terminá-lo mais rápido, ou lhe proporcionaria maior conforto ao realizá-lo. Cada uma, também, tinha o seu peso, e meu pai, forte como era, não conseguia carregar todas. Então, ele compilou uma tabela parecida com a sua, com um “valor” que cada ferramenta lhe proporcionaria, assim como seu peso.”

“Mas, meu caro, esse problema parece muito mais simples do que o meu: seu pai só precisava decidir entre levar uma ferramenta ou não, enquanto que eu preciso decidir entre quantos de cada tipo de item levar!” - disse Sulla.

“É verdade que o problema de meu falecido pai parece muito mais simples do que o seu. Mas toda pessoa bem educada, como tenho certeza ser o seu caso, oh conquistador de nações!, sabe que não se pode confiar nas aparências. Neste caso, ver que os problemas são essencialmente os mesmos é muito simples. Toda criança sabe que qualquer número natural pode ser decomposto como uma soma de potências de dois. Por exemplo,  $11 = 2^0 + 2^1 + 2^3$ . Além disso, essa decomposição é única, cada potência aparecendo no máximo uma vez. Como pode ver, então, o seu problema é um problema de decisão: decidir quais potências de 2 comporão as quantidades de cada tesouro que levará. Esse problema só tem o possível inconveniente de fazer uso de mais variáveis. Mas não muito mais:  $\lfloor \log_2(q_i) \rfloor$ , em que  $q_i$  é a quantidade a ser levada do item  $i$ .<sup>3</sup>”

“Mas esse problema adiciona complicações extras que não me parecem tão necessárias.” - disse Sulla.

“Não tanto assim, mas tudo bem. Não quero tomar muito mais tempo seu, então vou direto ao ponto. Um modelo mais geral ao problema que deseja resolver é o seguinte:

$$\begin{aligned} & \text{maximizar} \quad \sum_{i=1}^n q_i \mathcal{S}_i, \\ & \text{tal que} \quad \sum q_i p_i \leq C, \\ & 0 \leq q_i \leq l_i, q_i \text{ inteiro} \end{aligned}$$

---

<sup>3</sup>Estamos adicionando, aqui e mais adiante, notação um pouco mais moderna do que a que o velho Theon poderia ter usado, por conveniência ao exemplo.

Em que  $q_i$  é a quantidade a ser levada do item  $i$ ,  $p_i$  o peso do item  $i$  (em  $\mathcal{S}$ ),  $C$  a capacidade máxima a ser levada e  $l_i$  limite máximo do item  $i$  a ser levado (não podemos levar mais do que  $l_i$  do item  $i$ , talvez porque não existam mais do que  $l_i$  desses itens).

Esse é um problema de programação inteira. Existem muitas formas de resolver problemas de programação inteira. Se tiver algum tempo sobrando, eu poderia desenvolver uma outra forma com você. Será divertido e intrutivo.”

“Vá em frente.” - disse Sulla.

“Imagine que você sabe resolver esse problema para  $C' < C$  e para todo subconjunto dos itens (um subconjunto de  $\{1, \dots, n\}$ ) e tenha em mãos uma solução para ele. Se retirarmos dessa solução um item  $i$ , de peso  $p_i$ , deve ser claro que a solução restante é ótima para o problema cuja capacidade de  $C - p_i$  e conjunto de itens  $\{1, \dots, n\} - i$ . Isso indica que uma solução ótima goza de uma subestrutura ótima. Isto é, o retiro de itens dessa solução gera uma solução que é ótima para outro problema, de fato um subproblema, mais simples. Se temos a solução para algum desses subproblemas, só precisamos expandi-lo, considerando itens que não foram considerados. Para cada um desses itens, sabendo que a nova solução ótima  $op(i, c)$  para uma capacidade  $c$ , ao considerar a adição de um item  $i$ , é:

$$op(i, c) = \begin{cases} op(i-1, c), & \text{se } p_i > c \\ \max\{op(i-1, c), op(i-1, c-p_i) + \mathcal{S}_i\} & \text{se } p_i \leq c \end{cases}$$

Fazendo  $op(0, C) := 0$  e computando os valores de  $op(i, c)$  por essa recursão<sup>4</sup> nos dá o valor ótimo que o meu pai queria. Se quisermos resolver o seu problema, precisamos fazer apenas uma pequena alteração, como a seguinte:

$$op(i, c) = \max_{q_i} \{op(i-1, c - q_i p_i) + q_i \mathcal{S}_i \mid c \geq q_i p_i\}$$

Não deve ser difícil de ver como isso nos leva ao resultado desejado.”

“Oh, mas Theon! Devo ter me enganado ao achar que era um matemático, pois você me está parecendo mais um ator de teatro! Isso porque qualquer ator de teatro me daria uma resposta como essa, mas ela é longe de satisfatória. Mas é pior do que um ator de teatro, porque quem quer que fosse incubido de realizar a tarefa pelo método que você propôs ficaria um tempo intolerável fazendo contas, tempo esse que é, de fato, morto. E você seria o responsável pela morte.” - disse Sulla.

Theon, percebendo a gravidade da situação, disse o seguinte. “Essa formulação, como eu disse antes, tinha um caráter mais introdutório e instrutivo. Se quer uma mais séria, existem várias e em vários sabores. Uma delas em particular faz uso extenso de técnicas de programação inteira, que são desenvolvidas

---

<sup>4</sup>Conhecida como recursão de Bellman

a muitas décadas<sup>5</sup>. Fazendo uso dessas técnicas, podemos gerar o seguinte programa:

```
:- lib(eplex).

modelo(Pesos, Valores, Limitantes, Capacidade, Quantidades, Peso, Valor) :-
    ( foreach(Limitante,Limitantes), foreach(Quantidade,Quantidades) do
        Quantidade $:: 0..Limitante
    ),
    integers(Quantidades),
    Quantidades*Pesos $= Peso,
    Peso $=< Capacidade,
    Quantidades*Valores $= Valor.

resolve(Pesos, Valores, Limitantes, Capacidade, Quantidades, Peso, Valor) :-
    modelo(Pesos, Valores, Limitantes, Capacidade, Quantidades, Peso, Valor),
    optimize(max(Valor), Valor).

principal :-
    dados(Nomes, Pesos, Valores, Limitantes, Capacidade),
    resolve(Pesos, Valores, Limitantes, Capacidade, Quantidades, Peso, Valor),
    ( foreach(Quantidade,Quantidades), foreach(Nome,Nomes) do
        ( Quantidade >0 -> printf("%d of %w%n", [Quantidade,Nome]) ; true )
    ),
    writeln(peso = Peso ; valor = Valor).
```

Código 1: Conquistador Bárbaro Eplex

Podemos, analogamente, fazer uso da extensa pesquisa em otimização por restrições, com algo como o seguinte programa:

---

<sup>5</sup>Infelizmente, não há evidências de desenvolvimentos de programação inteira na época de Theon, mas a fala é mantida aqui por sua conveniência.

```

:- lib(ic).
:- lib(branch_and_bound).

model(Pesos, Valores, Limitantes, Capacidade, Quantidades, Peso, Valor) :-
    ( foreach(Limitante,Limitantes), foreach(Quantidade,Quantidades) do
        Quantidade $:: 0..Limitante
    ),
    integers(Quantidades),
    Quantidades * Pesos $= Peso,
    Peso $=< Capacidade,
    Quantidades * Valores $= Valor.

resolve(Pesos, Valores, Limitantes, Capacidade, Quantidades, Peso, Valor) :-
    model(Pesos, Valores, Limitantes, Capacidade, Quantidades, Peso, Valor),
    Custo #= -Valor,
    minimize(search(Quantidades, 0, largest, indomain_reverse_split, complete, []),
        Custo).

principal :-
    dados(Nomes, Pesos, Valores, Limitantes, Capacidade),
    resolve(Pesos, Valores, Limitantes, Capacidade, Quantidades, Peso, Valor),
    ( foreach(Quantidade,Quantidades), foreach(Nome,Nomes) do
        ( Quantidade > 0 -> printf("%d of %w%n", [Quantidade,Nome]) ;
          true )
    ),
    writeln(peso=Peso ; valor=Valor).

```

Código 2: Conquistador Bárbaro IC

Para saber qual dos métodos é o mais interessante, podemos simplesmente tomar uma pequena quantidade dos itens iniciais, fazer um teste, e ver qual resolvemos mais rápido.”

“Isso pode ser uma dificuldade, porque, como ainda não alcançamos todos os lugares de onde os itens serão subtraídos, só temos uma ideia muito vaga sobre quais itens existem e qual o valor de cada.” - Disse Sulla.

“Um problema facilmente remediado. Podemos fazer testes com valores aleatórios, como exemplificado aqui.” e, ao dizer isso, Theon primeiro explicou a notação que usaria e, depois, mostrou algo equivalente ao seguinte código a Sulla:

```

:- lib(ic_prop_test_util).

aleat_dados(dados([], [], [], [], _), 0).
aleat_dados(dados([Contador|Ls], [P|Ps], [V|Vs], [Q|Qs], _), Contador) :-
    random_int_between(1, 50, P),
    random_int_between(1, 100, V),
    random_int_between(1, 15, Q),
    Novo_Contador is Contador - 1,
    aleat_dados(dados(Ls, Ps, Vs, Qs, _), Novo_Contador).

inic_dados(dados(Nomes, Pesos, Valores, Qtd, _)) :-
    aleat_dados(dados(Nomes, Pesos, Valores, Qtd, _), 100).

dados(Nomes, Pesos, Valores, Qtd, 400) :-
    init_dados(dados(Nomes, Pesos, Valores, Qtd, 400)).

```

Código 3: Problemas Aleatórios

“Desta forma, se nos limitarmos a 50 segundos de computação, pela primeira versão do método<sup>6</sup>, podemos resolver o problema para 25 itens em um tempo médio de 0,01 segundos, enquanto que pela segunda versão do método<sup>7</sup>, para 14 itens, o tempo de resolução varia de 0,17 segundos a 34,23 segundos. Para 15 itens ou mais, não parece haver esperanças para o segundo método, enquanto que, pelo primeiro, problemas de 250.000 tipos de itens levam uma média de apenas 39 segundos para serem resolvidos.”

“Parece que, pelo menos por ora, nosso problema está resolvido.” - Disse Sulla.

“Sim, o seu problema está.”

---

<sup>6</sup>O programa 1.

<sup>7</sup>O programa 2.

## Leituras adicionais

- [1] Kellerer, Hans e Pferschy, Ulrich e Pisinger, David, “Knapsack Problems”, Springer
- [2] Plutarch, “Sulla”, traduzido por John Dryden
- [3] *ECL<sup>i</sup>PS<sup>e</sup>*, <http://www.eclipseclp.org/examples/>